


When we observe a collection of stars (galaxy, globular cluster, ...) the light we see at a particular location in the sky is the sum of the light of all the stars along that line of sight. The physical quantity of interest is the three dimensional distribution of stars. That is we have a problem of the sort:

what we observe = integral of what we are interested in.

and the problem is to invert this equation, expressing what we want (some how) in terms of what we can observe. Problems like this (called integral equations) are almost as common as differential equations. For example, X-ray absorption through a body is the sum of X-ray absorption of all of the parts of the body along the sight line. Computed Tomography (CT scans) do the inversion to display structures inside the body based on many line-of-sight X-ray absorption measurements.



The above displays the cross-section of a spherical object with a sight line penetrating the object; z is measured along the sight line with zero at the closest approach to the actual center. We assume a spherical distribution of stars, $\rho(r)$ and a measured light intensity $I(R)$ at some distance, R , from the 2D center of the object as seen in the sky.

$$I(R) = \int_{-\infty}^{+\infty} \rho(r) dz \quad (1)$$

Since $z^2 + R^2 = r^2$ we can change variables to r (so $zdz = r dr$), and at the same time use symmetry to argue that $-z$ contributes as much as $+z$

$$I(R) = 2 \int_R^{+\infty} \frac{\rho(r) r dr}{z} = 2 \int_R^{+\infty} \frac{\rho(r) r dr}{\sqrt{r^2 - R^2}} \quad (2)$$

In 1823 Niels Henrik Abel¹ published how to invert this problem:

$$\rho(r) = -\frac{1}{\pi r} \frac{d}{dr} \int_r^{+\infty} \frac{I(R) R dR}{\sqrt{R^2 - r^2}} = -\frac{1}{\pi} \int_r^{+\infty} \frac{I'(R) dR}{\sqrt{R^2 - r^2}} \quad (3)$$

It would be fun to prove this with the help of some complex mathematics, but I resist.

Instead let's use Mathematica to see the result in action. Note that the $\sqrt{R^2 - r^2}$ in the denominator with $R \rightarrow r$ will bother Mathematica, so some careful code will be required to get around Mathematica's (quite proper) concerns.

¹abelian is named for him

Begin by assuming a Gaussian distribution in 3D: $\rho = e^{-r^2}$, finding the resulting $I(R)$ and then inverting that. You should get what you started with!

```
rho[r_]=Exp[-r^2]
i[R_]=Simplify[2 Integrate[rho[r] r/Sqrt[r^2-R^2],{r,R,Infinity}],R>0]
```

Note: in Mathematica I'm using R and r to be different, which is probably a bad idea. Notice that a Gaussian distribution in 3D produces a Gaussian distribution on the 2D sky. Invert that result and report if the Abel integral gives you what you started with.

```
-1/Pi Integrate[D[i[R],R]/Sqrt[R^2-r^2],{R,r,Infinity}]
```

Consider now a sphere (radius 1) of constant density $\rho = 1$; since $\rho(r) = 0$ for $r > 1$ there is no need to integrate beyond $r = 1$. Again: check (and report) the results of inversion.

```
i[R_]=2 Integrate[ r/Sqrt[r^2-R^2],{r,R,1},GenerateConditions->False]
-1/Pi Integrate[D[i[R],R]/Sqrt[R^2-r^2],{R,r,1}]
```

Notice from the functional form of $I(R)$ that it goes to zero at the edge of the distribution.

Consider now a constant density shell of emission just in the range $.9 < r < 1$. We can reuse our above result if we think of ρ as a superposition of a $\rho = 1$ sphere of radius 1 and a $\rho = -1$ sphere of radius .9. We know how to do such spheres!

```
shell[R_]=If[R<1,i[R],0]
inside[R_]=If[R<.9,Evaluate[-2 Integrate[ r/Sqrt[r^2-R^2],{r,R,.9},
GenerateConditions->False]],0]
total[R_]=If[R<.9, inside[R]+shell[R],shell[R]]
Plot[total[R],{R,0,1.1},PlotRange->All]
```

The above plot is an important result: a shell of, say emitting gas, is 'limb bright' (limb refers to the edge of a sphere).

It turns out that, beyond the bulge, the density of stars in a spiral galaxy falls off exponentially:

```
rho[r_]=Exp[-r]
```

Find the resulting $I(R)$ and plot it together with $\rho(r)$

It turns out² that the observed brightness of an elliptical galaxy is very centrally peaked:

```
i[R_]=Exp[-7.669(R^(1/4)-1)]
Plot[i[r],{r,0,1},PlotRange->All]
```

²de Vaucouleur

This plot shows basically nothing at all. Instead results are usually plotted logarithmically as a function of $r^{1/4}$:

```
ParametricPlot[{x, 2.5 Log10[i[x^4]]}, {x, 0, 1}, AspectRatio->1]
```

Here the Abel integral has no algebraic solution, so we have Mathematica numerically integrate. Further the drop-off in intensity is so steep we can approximate infinity as 16!

```
rho[r_]:= -1/Pi NIntegrate[D[i[R],R]/Sqrt[R^2-r^2],{R,r,16}]
ParametricPlot[{x, 2.5 Log10[rho[x^4]]}, {x, 0, 1}, AspectRatio->1]
Plot[{rho[x], i[x]}, {x, .5, 4}, AspectRatio->1]
```

From these plots see that ρ is even more centrally peaked than $I(R)$. Remark: note the := in the above.

Homework

Provide a concise write up that includes the plots etc. underlined in the above text. If you are working together, feel free to submit one answer with several names attached. (Note: the definition of ‘working together’ is that any group member can describe what’s going on in the solution.)

Extra Credit: Can you prove the second equality in Eq. 3?